
Jörg Ackermann 

Elzenbergstraße 30E

08373 Niederlungwitz

firma@asoft-ackermann.de

TreeView-Control  -  eine kurze Praxiseinführung

Ein Strukturansicht-Steuerelement (TreeView) zeigt eine hierarchische Liste von Node-Objekten (Knoten) an, die aus einer Beschriftung und einer optionalen Bitmap bestehen. Ein Strukturansicht-Steuerelement wird in der Regel zum Anzeigen von Dokument-Überschriften, Indexeinträgen, Datei- und Verzeichnisstrukturen auf einem Datenträger oder zum Anzeigen beliebiger anderer Informationen verwendet werden, wo es nützlich ist diese als Hierarchie darzustellen

Benötigte Dateien

Die folgenden Dateien werden benötigt:

ODE VB 4 / 5    Comctl32.ocx     
Comcat.dll     
Comctl32.dep

VB 5 SP2 / 6
MSCOMCTL.OCX 
     Comcat.dll 
              MSCOMCTL.dep

Dep-Dateien werden zur Verwendung der Steuerelemente in der Entwicklungsphase benötigt und dürfen nicht distribuiert werden.

Unbedingt die Hinweise des EULA beachten.

PRIVATE
Versionen
VB5 Common Controls (SP2)         comctl32.dll. 

VB6 Common controls

     comctl32.dll 

VB6 Common controls (SP3)
     mscomctl.ocx. 

Das OCX ist nicht von der DLL abhängig. Es bietet erweiterten Komfort und neue Methoden /Funktionen z.B. zur Farbeinstellung.

Alle Versionen sind in Access verwendbar.

Hinweis zum Vertrieb
Das Strukturansicht-Steuerelement ist Bestandteil einer Gruppe von ActiveX-Steuerelementen, die sich in der Datei COMCTL32.OCX oder MSCOMCTL.OCX befinden. Sie können das Strukturansicht-Steuerelement in Ihrer Anwendung verwenden, indem Sie Ihrem Projekt die Dateien MSCOMCTL.OCX hinzufügen. Installieren Sie für einen Vertrieb Ihrer Anwendung die Datei MSCOMCTL.OCX im Microsoft Windows-Verzeichnis \System oder \System32.

Eigenschaften - Übersicht

Style-Property:
Es stehen zahlreiche Stile zum Gestalten der Darstellung des Steuerelements zur Verfügung. Node-Objekte können in einer von acht Kombinationen aus Texten, Bitmaps, Linien und Plus/Minus-Zeichen dargestellt werden.

PRIVATE
Constante
Wert
Beschreibung

tvwTextOnly
0
Nur Text wird gezeigt

tvwPictureText
1
Text und Grafik 

tvwPlusMinusText
2
Text mit Plus- oder Minus-Symbol

tvwPlusPictureText
3
Text mit Grafik und Plus- oder Minus-Symbol

tvwTreelinesText
4
Text und Linien

tvwTreelines-PictureText
5
Text mit Grafik und Linien

tvwTreelines-PlusMinusText
6
Text und Linien und Plus- oder Minus-Symbol

tvwTreelinesPlus-MinusPictureText
7
Text mit Grafik und Linien und Plus- oder Minus-Symbol
(Default)

MousePointer-Property:

Gibt einen Wert zurück oder legt einen Wert fest, der angibt, welche Art von Mauszeiger angezeigt wird, wenn sich die Maus zur Laufzeit über einem bestimmten Teil eines Objekts befindet.

PRIVATE
Konstante
Wert
Beschreibung

ccDefault
0
(Voreinstellung) Darstellung bestimmt durch das Objekt.

ccArrow
1
Pfeil.

ccCross
2
Kreuz, Kreuz-Zeiger.

ccIbeam
3
I-Cursor.

ccIcon
4
Symbol (kleines Quadrat innerhalb eines Quadrats.

ccSize
5
Größe (nach Norden, Süden, Osten und Westen zeigender Pfeil).

ccSizeNESW
6
Größe NO-SW (nach Nordost und Südwest zeigender Doppelpfeil).

ccSizeNS
7
Größe N S (nach Norden und Süden zeigender Doppelpfeil).

ccSizeNWSE
8
Größe NW SO (nach Nordwesten und Südosten zeigender Doppelpfeil)

ccSizeEW
9
Größe W O (nach Westen und Osten zeigender Doppelpfeil).

ccUpArrow
10
Nach oben gerichteter Pfeil.

ccHourglass
11
Sanduhr (Wartezustand).

ccNoDrop
12
Kein Ablegen möglich.

ccArrowHourglass
13
Pfeil und Sanduhr.

cc ArrowQuestion
14
Pfeil und Fragezeichen.

ccSizeAll
15
Größe für alle Richtungen ändern.

ccCustom
99
Benutzerdefiniertes Symbol (angegeben durch MouseIcon-Eigenschaft)

LineStyle-Property :

Falls Linien gezeigt werden.bestimmt die LineStyle-Property (0-tvwTreeLines) oder (1-tvwRootLines), ob Linien vor oder nach dem Root-Knoten beginnen.

Indentation 

Entfernung des Kind-Knotens vom Perent-Knoten

LabelEdit 

Standardmäßig kann die Beschriftung eines Nodes durch Klick geändert werden 

( 0-tvwAutomatic) Falls auf manual 

(1-tvwManual) gesetzt, kann die beschreibung nur durch die StartLabelEdit-Methode gesetzt werden. 

PathSeparator

Gibt die als Trennung in Pfadangaben dienende Zeichenfolge zurück oder legt sie fest. Pfadangaben werden von der FullPath-Eigenschaft zurückgegeben. Standard: \ (Backslash)

ImageList 

Gibt ein Abbildungsliste-Steuerelement (ImageList) zurück oder legt ein Abbildungsliste-Steuerelement fest (sofern vorhanden) Damit für das Steuerelement die ImageList-Eigenschaft verwendet werden kann, müssen Sie ein Abbildungsliste-Steuerelement im Formular positionieren.

BorderStyle

Gibt den Rahmenstil für ein Objekt zurück oder legt ihn fest

PRIVATE
Constante
Wert
Beschreibung

ccNone
0
Kein

ccFixedSingle
1
rahmen

Appearance

Gibt zurück oder legt fest, welche Stile für die Ausgabe von Steuerelementen in einem MDIForm- oder Form-Objekt zur Entwurfszeit verwendet werden. Zur Laufzeit schreibgeschützt

PRIVATE
Constante
Wert
Beschreibung

ccFlat
0
flach

cc3D
1
vertieft 

OLEDragMode

Gibt zurück oder legt fest, ob die Komponente oder der Programmierer eine OLE-Drag/Drop-Operation übernimmt.

PRIVATE
Konstante 
Wert 
Beschreibung 

ccOLEDragManual 
0 
Manuell. Der Programmierer übernimmt alle OLE-Drag/Drop-Operationen. 

ccOLEDragAutomatic 
1 
Automatisch. Die Komponente übernimmt alle OLE-Drag/Drop-Operationen. 

OLEDropMode

Gibt zurück oder legt fest, wie eine Zielkomponente Ablege-Operationen durchführt

PRIVATE
Konstante 
Wert 
Beschreibung 

ccOLEDropNone 


0 
Keine. Die Zielkomponente akzeptiert keine OLE-Drop-Operationen und zeigt einen Cursor an, der darstellt, daß an dieser Position kein Objekt abgelegt werden kann. 

ccOLEDropManual 


1 
Manuell. Die Zielkomponente löst die OLE-Drop-Ereignisse aus und ermöglicht so dem Programmierer die Durchführung der OLE-Drop-Operation im Code. 

CcOLEDropAutomatic 


2 
Automatisch. Die Zielkomponente akzeptiert automatisch OLE-Drop-Operationen, falls das DataObject-Objekt Daten in einem Format enthält, das es erkennt. Keine Maus- oder OLE-Drag/Drop-Ereignisse treten auf, wenn die OLEDropMode-Eigenschaft auf ccOLEDropAutomatic gesetzt wird. 

Folgende weitere Properties erlauben nur True/False-Werte

HideSelection

Gibt einen Wert zurück, der bestimmt, ob der ausgewählte Text hervorgehoben angezeigt wird, wenn ein Steuerelement den Fokus abgibt. 

Sorted

Das Treeview selbst kann manuell nicht sortiert werden. Wenn die Sorted-property auf True gesetzt ist, werden alle Kindknoten innerhalb des Parents alfabetisch sortiert. Diese Eigenschaft ist global und betrifft alle Knoten des Controls. Jeder Node hat aber selbst eine Sorted-Property, die bestimmt, ob seine Kindknoten sortiert werden. 

FullRowSelect

Gibt einen Wert zurück bzw. legt einen Wert fest, der bestimmt, ob die ganze Zeile ausgewählt ist.

Enabled

Gibt einen Wert zurück oder legt einen Wert fest, der bestimmt, ob ein Formular oder ein Steuerelement auf vom Benutzer ausgelöste Ereignisse reagieren kann.

Checkboxes

Gibt einen Wert zurück bzw. legt einen Wert fest, der bestimmt, ob Kontrollkästchen angezeigt werden.
SingleSel

Gibt einen Wert zurück bzw. legt einen Wert fest, der bestimmt, ob ein Element beim Auswählen erweitert wird.

Scroll

Gibt einen Wert zurück bzw. legt einen Wert fest, der bestimmt, ob Bildlaufleisten angezeigt werden.

HotTracking

Gibt einen Wert zurück, der bestimmt, ob maussensitive Marierung aktiv ist.

Alles deht sich um NODES...

Ein Node-Objekt ist ein Element eines TreeView, das Bilder und Text enthalten kann.  Eine Nodes-Auflistung enthält eines oder mehrere Node-Objekte. Jedes Element in einem Treeview ist ein Node-Objekt.

Node-Objekte können sowohl Text als auch Bilder enthalten. Jedoch müssen Sie, um Bilder verwenden zu können, ein Abbildungsliste-Steuerelement (ImageList) mit Hilfe der ImageList-Eigenschaft zuordnen.

Das Struckturansicht-Steuerelement verwendet zum Speichern von in Node-Objekten dargestellten Bitmaps und Symbolen das Abbildungsliste-Steuerelement (ImageList), das durch die ImageList-Eigenschaft angegeben wird. Ein Strukturansicht-Steuerelement kann jeweils immer nur eine Abbildungsliste verwenden. 

Dies bedeutet, daß neben jedem Element in einem Strukturansicht-Steuerelement ein Bild gleicher Größe angezeigt wird, wenn die Style-Eigenschaft des betreffenden Strukturansicht-Steuerelements auf einen Stil festgelegt wird, der das Anzeigen von Bildern angibt.

Die in einem Node-Objekt enthaltenen Bilder können sich je nach Zustand des Objekts ändern. So kann beispielsweise ein ausgewähltes Node-Objekt ein anderes Bild anzeigen als ein nicht ausgewähltes Node-Objekt, wenn Sie die SelectedImage-Eigenschaft auf ein Bild aus dem zugeordneten Abbildungsliste-Steuerelement festlegen.

Wichtigste Node-Eigenschaften: 

Count, Image, Index, Item, Key, Tag, Text, Visible, Expanded, Selected

Nodes hinzufügen:    Add-Methode

Syntax

Objekt.Add(relativ, Beziehung, Schlüssel, Text, Bild, ausgewähltesBild)

Wenn in relativ kein Node-Objekt angegeben ist, dann wird das neue Node-Objekt in der höchsten Hierarchie-Ebene an letzter Position plaziert

Die Einstellungen für Beziehung sind:

PRIVATE
Konstante
Wert
Beschreibung

tvwFirst
0
Erste. Das Node-Objekt wird vor alle anderen Node-Objekte auf derselben Stufe plaziert, die der Stufe des in relativ angegebenen Node-Objekts entspricht.

tvwLast
1
Letzte. Das Node-Objekt wird hinter alle anderen Node-Objekte auf derselben Stufe plaziert, die der Stufe des in relativ angegebenen Node-Objekts entspricht. Jedes später hjinzugefügte Node-Objekt kann hinter einem als letztes hinzugefügtem plaziert werden.

tvwNext
2
(Voreinstellung) Nächste. Das Node-Objekt wird hinter dem in relativ angegeben Node-Objekt plaziert.

tvwPrevious
3
Das Node-Objekt wird vor dem in relativ angegebenen Node-Objekt plaziert.

tvwChild
4
Untergeordnetes Node-Objekt. Das Node-Objekt ist dem in relativ angegebenen Node-Objekt untergeordnet.

Wenn der Beziehungs-Parameter auf 0-tvwFirst, 1-tvwLast, 2-tvwNext, oder 3-tvwPrevious gesetzt wird, wird der neue Node in der selben Ebene platziert wie der Elternknoten. Wenn auf 4-tvwChild gesetzt wird, ist der neue Node ein Kind-Knoten

Node.Key muß ein String sein und muß einen alphanumerischen Wert enthalten. 

( z.B. "a456"  unzulässig:  "3426")      siehe auch Probleme...

Wenn ein Node-Objekt hinzugefügt wird, dann erhält es eine Index-Nummer, die in der Index-Eigenschaft des Node-Objekts gespeichert wird. Der Wert des aktuellsten Elements entspricht dem Wert der Count-Eigenschaft der Nodes-Auflistung.

Da die Add-Methode einen Verweis auf das neu erstellte Node-Objekt als Ergebnis zurückgibt, ist es am einfachsten, diesen Verweis zum Festlegen der Eigenschaften eines neuen Node-Objekts zu verwenden. 

Dim nodX As Node

Set nodX = tvwTree.Nodes.Add(,,"r1","Root")

nodX.Expanded = True

nodX.Image = 1

nodX.ExpandedImage = 2

Set nodX = tvwTree.Nodes.Add("r1",tvwChild,"k1","Knoten1")

nodX.Selected = True

nodX.Image = 1

nodX.ExpandedImage = 2
Nodes entfernen:

Remove-Methode

Syntax
Objekt.Remove Index
tvwTree.Nodes.Remove tvwTree.SelectedItem.Index

tvwTree.Nodes.Remove tvwTree.Nodes(4)

tvwTree.Nodes.Remove 4

Selektieren eines Nodes 

SelectedItem (Treeview-Control), Selected (Node-Object) 

Die SelectedItem-Property des Treeview gibt das gerade ausgewählte Node object zurück. Die Selected -Property eines Node-Objects gibt einen boolschen Wert zurück, der festlegt, ob das Objekt gerade selektiert ist.

AktuellerText = tvwTree.SelectedItem.Text
AktuellerText = tvwTree.Nodes(tvwTree.SelectedItem.Index)

tvwTree.Nodes(3).Selected = True

IsNodeSelected  = tvwTree.Nodes(3).Selected

Node-Navigation

Parent 

Die Parent-Property eines Nodes gibt den höheren(Eltern)-Knoten in der Hirarchie der Knoten zurück. Alle Versuche, einen Knoten als seinen eigenen Elternknoten zu definieren werden ein Loch im Raum-Zeit-Kontinuum hinterlassen (or cause an error). 

Root 

Gibt den höchsten Node in der Hierarchie zurück. 

FullPath 

Gibt den kompletten Pfad zum aktuellen Node zurück 

Children, Child 

Die Children-Property gibt die Anzahl der Kind-Knoten zurück. Die Child-Property gibt den ersten Kind-Knoten zurück. Ist kein Kind-Knoten vorhanden, wird ein Fehler erzeugt, also vorher die Childs-Property testen... 

FirstSibling, LastSibling, Next, Previous 

Geben Knoten der selben Ebene zurück und sind selbsterklärend.

Private Sub tvwTree_DblClick()

    Dim nodItem As Node

    If Me.tvwTree.SelectedItem.Children > 0 Then

        'gibt es Children

        Set nodItem = Me.tvwTree.SelectedItem.Child

        'ersten Kind-Knoten ermitteln

        Do

            Debug.Print nodItem.Text

            If nodItem = nodItem.LastSibling Then

                'letzten Knoten ermitteln

                Exit Do

            Else

                'nächsten Knoten ermitteln

                Set nodItem = nodItem.Next

            End If

        Loop

    End If

End Sub

Das Bild zeigt ein Treeview mit sechs Node-Objects. 

Einige Beziehungen:

· Node1.Child = Node2 

· Node2.FirstSibling = Node2 

· Node2.Next = Node3 

· Node2.LastSibling = Node6 

· Node6.Previous = Node3 

Methoden:

GetVisibleCount

Gibt die Anzahl der Node-Objekte zurück, die in den Innenbereich eines Strukturansicht-Steuerelements passen.
HitTest 

Gibt einen Verweis auf ein ListItem- oder Node-Objekt zurück, das an den Koordinaten x und y positioniert ist. Wird mit Drag/Drop-Operationen verwendet.. 

OLEDrag
Startet ein OLE-Drag/Drop-Ereignis mit dem angegebenen Steuerelement als Quelle.

Refresh

Erzwingt ein vollständiges Neuzeichnen

StartLabelEdit 

Wenn die StartLabelEdit-Methode aufgerufen wird, wird der Beschreibungstext des Nodes editierbar. Dies ist überflüssig, wenn die LabelEdit-Property auf 0-tvwAutomatic gesetzt ist.

Ereignisprozeduren:

Event AfterLabelEdit(Cancel As Integer, NewString As String)

Tritt auf, nachdem ein Benutzer die Bezeichnung des momentan ausgewählten Node- oder ListItem-Objekts bearbeitet hat.

Event BeforeLabelEdit(Cancel As Integer)

Tritt auf, wenn ein Benutzer versucht, die Bezeichnung des momentan ausgewählten Node- oder ListItem-Objekts zu bearbeiten.

Event Click()

Tritt auf, wenn der Benutzer eine Maustaste über einem Objekt drückt und wieder losläßt.

Event DblClick()

Tritt auf, wenn man eine Maustaste drückt und wieder losläßt und nochmals über einem Objekt drückt und wieder losläßt.

Event Expand(Node As Node)



Event Collapse(Node As Node)

Tritt auf, wenn ein Node-Objekt in einem Strukturansicht-Steuerelement geöffnet/geschlossen wird.

Das Expand-Ereignis kann auch zur Umkehr dieser Aktion benutzt werden: 

Private Sub tvwTree_Expand(ByVal Node As Object)

    If Node.Tag = "Nicht Aufkappen" Then

        'wieder zuklappen

        Node.Expanded = False

    End If

End Sub

Event KeyDown(KeyCode As Integer, Shift As Integer)

Tritt auf, wenn der Benutzer eine Taste drückt, während ein Objekt den Fokus hat.

Event KeyPress(KeyAscii As Integer)

Tritt auf, wenn der Benutzer eine ANSI-Taste drückt und losläßt.

Event KeyUp(KeyCode As Integer, Shift As Integer)

Tritt auf, wenn der Benutzer eine Taste losläßt, während ein Objekt den Fokus hat.

Event MouseDown(Button As Integer, Shift As Integer, x As OLE_XPOS_PIXELS, _
                                 y As OLE_YPOS_PIXELS)

Tritt auf, wenn der Benutzer die Maustaste drückt, während ein Objekt den Fokus hat.

Event MouseMove(Button As Integer, Shift As Integer, x As OLE_XPOS_PIXELS, _
                              y As OLE_YPOS_PIXELS)

Tritt auf, wenn der Benutzer die Maus bewegt.

Event MouseUp(Button As Integer, Shift As Integer, x As OLE_XPOS_PIXELS, _
                            y As OLE_YPOS_PIXELS)

Tritt auf, wenn der Benutzer die Maustaste losläßt, während ein Objekt den Fokus hat.

Event NodeCheck(Node As Node)

Tritt auf, wenn Checkboxes = True und ein Node-Objekt aktiviert/deaktiviert ist.

Event NodeClick(Node As Node)

Tritt auf, wenn auf ein Node-Objekt geklickt wird. Der angeklickte Node wird dabei als Parameter übergeben. 

Private Sub tvwTree_NodeClick(ByVal Node As Object)

    If Node.Tag = "Meldung" Then

        'Meldung zeigen

        Msgbox "NodeKey: " & Node.Key & "  NodeIndex: " & _

                Node.Index

    End If

End Sub

Event OLECompleteDrag(Effect As Long)
Event OLEDragDrop(Data As DataObject, Effect As Long, Button As Integer, Shift As 

                                   Integer, x As Single, y As Single)

Event OLEDragOver(Data As DataObject, Effect As Long, Button As Integer, Shift As                 

                                  Integer, x As Single, y As Single, State As Integer)

Event OLEGiveFeedback(Effect As Long, DefaultCursors As Boolean)

Event OLESetData(Data As DataObject, DataFormat As Integer)

Event OLEStartDrag(Data As DataObject, AllowedEffects As Long)

Ereignisse zur Drag-Drop Funktionalität

Auffangbare Fehler für die Windows-Standardsteuerelemente (MSComCtlLib - MSComctl.ocx)

In der folgenden Tabelle sind die auffangbaren Fehler und Konstanten für die Windows-Standardsteuerelemente in der Datei MSComctl.OCX aufgeführt.

PRIVATE
Konstante
Wert
Beschreibung

CcInvalidProcedureCAll
5
Ungültiger Prozeduraufruf

CcOutofMemory
7
Nicht genügend Speicher. Die Operation konnte nicht genügend Speicher reservieren.

CcTypeMismatch
13
Typen unverträglich. Eines der Argumente konnte nicht in den richtigen Datentyp konvertiert werden.

CcObjectVariableNotSet
91
Objektvariable oder With-Blockvariable nicht festgelegt.

CcInvalidPropertyValue
380
Unzulässiger Eigenschaftswert



Einer Eigenschaft wurde ein Wert zugewiesen, der außerhalb des Bereichs der erlaubten Werte liegt.

CcSetNotSupportedAtRuntime
382
Eigenschaft kann zur Laufzeit nicht gesetzt werden.

CcSetNotSupported
383
Eigenschaft ist schreibgeschützt.

CcSetNotPermitted
387
Eigenschaft kann für dieses Steuerelement nicht gesetzt werden.

CcGetNotSupported
394
Eigenschaft kann nur gesetzt werden

CcInvalidObjectUse
425
Unzulässige Verwendung eines Objekts

CcWrongClipboardFormat
461
Zwischenablageformat ungültig

CcInvalidPicture
481
Ungültiges Bild

CcDataObjectLocked
672
Formatliste für DataObject-Objekt kann nicht außerhalb des OLEStartDrag-Ereignisses gelöscht oder erweitert werden

CcExpectedAnArgument
673
Mindestens ein Argument erwartet.

CcRecursiveOleDrag
674
Ungültiger, rekursiver Aufruf von OLE-Drag-Drop

CcFormatNotByteArray
675
Nichtunterstützte OLE-Ziehen/Ablegen-Formate werden mit SetData verwendet und erfordern Datenfelddaten vom Typ Byte. GetData gibt möglicherweise mehr Bytes zurück als mit SetData übergeben wurden.

CcDataNotSetForFormat
676
Angeforderte Daten wurden dem DataObject-Objekt während des OLESetData-Ereignisses nicht zur Verfügung gestellt.

CcInvalidSafeModeProcCall
680
Ungültiger Prozeduraufruf in Sicherheitsmodus.

CcIndexOutOfBounds
35600
Indexgrenze überschritten

ccElemNotFound
35601
Element nicht gefunden

ccNonUniqueKey
35602
Schlüssel ist in Auflistung nicht eindeutig

ccInvalidKey
35603
Ungültiger Schlüssel

ccCol1MustBeLeftAligned
35604
Wenn die View-Eigenschaft des Listenansicht-Steuerelements (ListView) auf 3 (lvwReport) gesetzt wird, kann die Spalte ganz links (Spalte 1) nur links ausgerichtet werden. Jeder Versuch, einen anderen Wert für die Ausrichtung zu setzen, führt zu diesem Fehler.

ccElemNotPartOfCollection
35605
Das Steuerelement dieses Elements wurde gelöscht.

ccCollectionChangedDuringEnum
35606
Die Auflistung des Steuerelements wurde modifiziert.

ccMissingRequiredArg
35607
Erforderliches Argument fehlt.

ccBadObjectReference
35610
Objekt ungültig

ccReadOnlyIfHasImages
35611
Die Eigenschaft ist schreibgeschützt, wenn die Abbildungsliste Abbildungen enthält.

ccImageListMustBeInitialized
35613
ImageList muß initialisiert werden, bevor sie verwendet werden kann.

ccWouldIntroduceCycle
35614
Dadurch würde ein Zyklus eingeführt.

ccNotSameSize
35615
Alle Abbildungen in der Liste müssen die gleiche Größe haben.

ccMaxPanelsExceeded
35616
Maximale Anzahl an Grundflächen überschritten

ccImageListLocked
35617
Abbildungsliste kann nicht verändert werden, während ein anderes Steuerelement daran gebunden ist.

ccMaxButtonsExceeded
35619
Maximale Anzahl an Schaltflächen überschritten

ccCircularReference
35700
Zirkelbezug von Objekten ist nicht zulässig.

Demonstration anhand einer einfachen Beispiel-Applikation  -

Der TreeView-Assistent

Um den Einsatz des Treeviews zu demonstrieren, habe ich als Beispiel einen Assistenten in Access erstellt, der es ermöglicht, komfortabel visuell Webseiten mit einem Java-Treeview zu erstellen.

Folgende Anforderungen sollte der Assistent erfüllen:

· Komfortable Visuelle TreeView-Gestaltung

· automatische Erstellung einer Web-Seite mit Java-Tree wahlweise als Eizelseite oder als Frameseite mit zwei oder drei Frames.

· Export in das Veröffentlichungsverzeichnis (lokales Netz)

· Freie Wahl des Exportverzeichnisses sowie der HTML-Dateinamen / Frame-IDs

· Zuweisung von verschiedenen Bildern für geöffnete/geschlossene Knoten
· automatische Erstellung der in der URL angegebenen Webseiten als Grundlage für weitere Gestaltung ( leere Seiten, aber keine toten lokalen Links )

· Speicherung der erstellten TreeViews für spätere Änderungen / Neuerstellung

· Nach Veränderung der Eigenschaften im rechten Teil wird das Tree sofort entsprechend aktualisiert. ( Wysiwyg )

Aufbau:

Im linken Teil wird das Treeview erstellt, im rechten Teil werden die Eigenschaften der späteren Java-Nodes festgelegt. 

Im Form wird zusätzlich zu den sichtbaren Steuerelementen ein Image-Control (imgTree) plaziert, das alle verwendeten Bilder (GIFs) enthält.

Eigenschaften des Trees:


Style: 

7-tvwTreelinesPlusMinusPictureText 
HideSelection:
False


LineStyle:
1-tvwRootLines



HotTracking:
True


Indentation:
200


ImageList:
imgTree

Einige Topics der Realisierung:

Im FormLoad-Ereignis wird ein Standard Tree mit einem Root-Node sowie einem Dokument (Home) erstellt, sowie die entsprechenden Eigenschaften im rechten Teil voreingestellt.

Auszugsweise dazu der relevante Code zur ersten Füllung des Trees, der auch beim Klicken des Neu-Buttons ausgeführt wird:

Private Sub Form_Open(Cancel As Integer)

    FillTree

End Sub

Sub FillTree(Optional FileID As Long)

Dim nodX As Node, strTagRoot As String, strTagFirst As String

strTagRoot = "0;0;1;welt2___.gif;welt2___.gif;" & _

"A-Soft Online;http://www.asoft-ackermann.de;_top"

strTagFirst = "1;2;1;doc_html.gif;doc_html.gif;Home;" & _

DLookup("txtMain", "tblOptions") & ";" & DLookup("IDMain", "tblOptions")

With tvwTree

   'erst mal alles leeren

   If .Nodes.count > 0 Then .Nodes.Clear

   'Root anlegen

   Set nodX = .Nodes.Add(, , "r0001", "A-Soft Online")

   nodX.Image = "welt2___.gif"

   nodX.ExpandedImage = "welt2___.gif"

   nodX.Expanded = True

   nodX.Tag = strTagRoot

   '1.Dokument anlegen

   Set nodX = .Nodes.Add("r0001", tvwChild, "d0101", "Home")

   nodX.Image = "doc_html.gif"

   nodX.ExpandedImage = "doc_html.gif"

   nodX.Tag = strTagFirst

End With

'Root markieren

 tvwTree.SelectedItem = tvwTree.Nodes(1)

'Eigenschaften füllen (Parsen des strTagRoot)

 FillForm tvwTree.SelectedItem.Index

End Sub
Als Key wird jeweis verwendet: 

· "r" für Root, "o" für Ordner, "d" für Dokument

· Ebene des Knotens im Format "00" nullbasierend

· Position des Knotens in der Ebene fortlaufend im Format "00" einsbasierend

So daß sich für den Root-Knoten als Key "r0001" und für den ersten Dokument-Knoten "d0101" ergibt.

In der Tag-Property befindet sich der String zur Formatierung der Nodes in ähnlicher Form, wie er später vom Java-Script benötigt wird. Er wird geparst und z.B. im NodeClick-Ereignis in die Steuerelemente im rechten Teil des programmfensters geladen.

"0;0;1;welt2.gif;welt2.gif;A-Soft Online;http://www.app.de;_top"

Bedeutung:  ( Ebene ; Node-Art ; Style ; Bild_zu ; Bild_auf ; Text ; Url ; Zielframe )

Um einen Ordner zu einem Node hinzuzufügen wird folgender Code verwendet (Auszug):

Private Sub btnAddOrdner_Click()

Dim nodX As Node, nKey As String, nEbene As Integer, nAnz As Long

Dim strTag As String

    nKey = tvwTree.SelectedItem.Key      'aktueller Key

    nEbene = CInt(Mid(nKey, 2, 2)) + 1   'Ebene ermitteln u. +1

    nAnz = GetNextKey(nEbene)            'nächsten freien Key d. Ebene erm.

                                         'siehe unten...

    strTag = nEbene & ";1;0;ord___zu.gif;ord__auf.gif;Neuer Ordner;;"

    'Neuer Node

    Set nodX = tvwTree.Nodes.Add(nKey, tvwChild, "o" & _

    Format$(nEbene, "00") & Format$(nAnz, "00"), "Neuer Ordner")

    nodX.Image = "ord___zu.gif"

    nodX.ExpandedImage = "ord__auf.gif"

    nodX.Tag = strTag

    'eingefügten Node markieren

    tvwTree.SelectedItem = tvwTree.Nodes(nodX.Index)

    'Eigenschaften in rechten Teil laden

    FillForm tvwTree.SelectedItem.Index

    'Kennzeichen für Speicheranfrage setzen

    SetUnsaved

End Sub

Das Hinzufügen eines Dokuments erfolgt in ähnlicher Form.

Die oben verwendete Function GetNextKey ermittelt den nächsten freien Wert für die Key-Eigenschaft:

Dazu wird die Nodes-Auflistung durchlaufen und der höchste Key in der gleichen Ebene ermittelt.

Function GetNextKey(nEbene As Integer) As Integer

Dim I As Integer, lngKey As Long

    GetNextKey = 0

    For I = 1 To tvwTree.Nodes.count

        lngKey = Mid(tvwTree.Nodes(I).Key, 4)

        If CInt(Mid(tvwTree.Nodes(I).Key, 2, 2)) = nEbene Then

            If lngKey > GetNextKey Then GetNextKey = lngKey

        End If

    Next I

    GetNextKey = GetNextKey + 1

End Function
Synchronisation nach Änderung einer Treeview - Eigenschaft:

Private Sub tvwTree_AfterLabelEdit(Cancel As Integer, NewString As String)

    txtDescription = NewString

    btnSaveNode_Click False

End Sub

Oder:

Private Sub tvwTree_Expand(ByVal Node As Object)

    FillForm Node.Index

End Sub

Oder:

Private Sub tvwTree_NodeClick(ByVal Node As Object)

    FillForm Node.Index

End Sub

Und in die andere Richtung:

Private Sub optStyle_AfterUpdate()

    tvwTree.SelectedItem.Expanded = CBool(optStyle)

    btnSaveNode_Click True

End Sub

Ereignisprozedur des 'Speichern'-Buttons:   Tag zusammensetzen

Private Sub btnSaveNode_Click(Optional opt As Boolean)

Dim nodX As Node

 'Tag neu zusammenfieseln und in Statuszeile schreiben

 tTag = CInt(Mid(tvwTree.SelectedItem.Key, 2, 2)) & ";" & CStr(optTyp) & _          

       ";" & CStr(optStyle) & ";" & txtBildClosed & ";" & txtBildOpened & _

       ";" & txtDescription & ";" & txtURL & ";" & txtUserFrame

 If opt Then   'ggf. Tree aktualisieren

   'wenn ein Knoten markiert ist

   If Not tvwTree.SelectedItem Is Nothing Then

     'Werte in markierten Knoten speichern

     Set nodX = tvwTree.Nodes(tvwTree.SelectedItem.Index)

     nodX.Text = txtDescription

     nodX.Image = CStr(txtBildClosed)

     nodX.ExpandedImage = CStr(txtBildOpened)

     nodX.Tag = tTag

     'Kennzeichen Speicheranfrage setzen

     SetUnsaved

   End If

  End If

End Sub

Um die Bilder komfortabel auswählen zu können, wurde ein Form mit einem ListView erstellt, das eine Anzeige der Bilder und deren Selektion ermöglicht.

Die Grafiken werden wiederum in einem Image-Control verwaltet, das an das ListView über die ImageList-Property gebunden ist.

Die Speicherung der Listview-Daten erfolgt in zwei simplen Tabellen:

Es wird lediglich in tblFiles eine Beschreibung der "Tree-Datei" und ein TimeStamp benötigt.

In der Detailtabelle tblFileDetails wird der Key, der zur Ermittlung der Ebene sowie zur Sortierung benötigt wird, sowie der Script-String ( Tag des Nodes ) abgelegt.

Nun ist es möglich, Struktur und Inhalt des Trees zu speichern und bei Bedarf wieder zu laden.

Prozeduren können Sie der Access-MDB ( Treeassi ) entnehmen, die sich auf der AEK-CD befindet.

Erstellung der HTML-Dateien und des Java-Scripts:


Vor dem Export des Scripts werden die Einstellungen des Treeviews in die o.g. Tabellen gespeichert. Anschließend werden Programmeinstellungen zum Export aus den entsprechenden Tabellen gelesen.

Programmeinstellungen werden folgendermaßen festgelegt:

Anschließend werden aus der Tabelle tblFileDetails die entsprechenden TreeView-Daten gelesen und daraus die Java-Scriptseite erstellt. Dieses wird wie auch die HTML-Seiten direkt in das Export-Verzeichnis geschreiben. Anschließend wird der Browser mit den erstellten Dateien gestartet.

Da es uns bei diesem Beispiel vorrangig um die Verwendung des Treeviews geht, will ich hier nicht 

näher auf diesen Programmteil eingehen.

Kurzer Code-Auszug zur Verdeutlichung des Prinzips:

------------Snip-----------------------------------

    Dat1 = FreeFile

    Open Dateiout For Output As Dat1

    strScript = "<html>" & vbCrLf

    strScript = strScript & "<head>" & vbCrLf

    strScript = strScript & "<title>Kopf</title>" & vbCrLf

    strScript = strScript & "<base target=" & Chr(34) & fMainid & _

                             Chr(34) & ">" & vbCrLf

    strScript = strScript & "</head>" & vbCrLf

    strScript = strScript & "" & vbCrLf

    strScript = strScript & "<body bgcolor=" & Chr(34) & "#FFFFFF" & _

                             Chr(34) & ">" & vbCrLf

    strScript = strScript & "</body>" & vbCrLf

    strScript = strScript & "</html>"

    Print #Dat1, strScript; 

--------------Snip-----------------

Geschafft !!!

Am Ende sollte das Ziel unser Bemühungen sichtbar sein - eine etwas spartanische, aber voll funktionstüchtige Frameseite mit einem fehlerlos arbeitendem Java-Tree.

So oder ähnlich sollte die Vorschau im Browser aussehen:

Erweiterte Code-Beispiele 1:

Farbenspiele / Einstellungen mit dem Treeview per API-Calls

Private Declare Function SendMessage Lib "user32" Alias _

"SendMessageA" (ByVal hWnd As Long, ByVal wMsg As Long, _

ByVal wParam As Long, lParam As Long) As Long
Private Declare Function GetWindowLong Lib "user32" Alias _

"GetWindowLongA" (ByVal hWnd As Long, ByVal nIndex As Long) As Long
Private Declare Function SetWindowLong Lib "user32" Alias _

"SetWindowLongA" (ByVal hWnd As Long, ByVal nIndex As Long, _

ByVal dwNewLong As Long) As Long
Private Const GWL_STYLE = -16&

Private Const TVM_SETBKCOLOR = 4381&

Private Const TVM_GETBKCOLOR = 4383&

Private Const TVS_HASLINES = 2&

Private Const TV_FIRST As Long = &H1100

Private Const TVM_GETTEXTCOLOR As Long = (TV_FIRST + 32)

Private Const TVM_SETTEXTCOLOR As Long = (TV_FIRST + 30)

Private Const TVS_CHECKBOXES = &H100

Private Const TVS_TRACKSELECT = &H200

Private Sub Form_Load()

Dim mNode As Node

Dim i As Integer
'Add the primary node
Set mNode = tvwTree.Nodes.Add()

mNode.Text = "Primary Node"

For i = 1 To 5


Set mNode = tvwTree.Nodes.Add(1, tvwChild)


mNode.Text = "Child " & i

Next i

'change the background
Call SendMessage(tvwTree.hWnd, TVM_SETBKCOLOR, 0, ByVal RGB(255, 204, 0))

'change the text colour
Call SendMessage(tvwTree.hWnd, TVM_SETTEXTCOLOR, 0, ByVal RGB(0, 127, 0)) 

'Add the track selection
Call SetTreeViewAttrib(tvwTree, TVS_TRACKSELECT)

'Add the checkboxes
Call SetTreeViewAttrib(tvwTree, TVS_CHECKBOXES)

End Sub

Private Sub SetTreeViewAttrib(c As TreeView, ByVal Attrib As Long)

    Const GWL_STYLE As Long = -16

    Dim rStyle As Long
    rStyle = GetWindowLong(c.hWnd, GWL_STYLE)

    rStyle = rStyle Or Attrib

    Call SetWindowLong(c.hWnd, GWL_STYLE, rStyle)

End Sub

Erweiterte Code-Beispiele 2:

Node-Text auf 50 Zeichen beschränken

This tip was provided by Chris Eastwood (http://www.codeguru.com/vb/) This tip uses the SendMessage API call with the TVM_GETEDITCONTROL and the EM_LIMITTEXT contants to set the length of a nodes text to 50 characters
PRIVATE
Option Explicit

Private Declare Function SendMessage Lib "user32" Alias _

"SendMessageA" (ByVal hwnd As Long, ByVal wMsg As Long, _

ByVal wParam As Long, lParam As Any) As Long

Private Const TV_FIRST = &H1100

Private Const TVM_GETEDITCONTROL = TV_FIRST + 15

Private Const EM_LIMITTEXT = &HC5 

Private Sub SetTextLen(tvw As Treeview, intSize As Integer)

    Dim lngHwnd As Long   

    lngHwnd = SendMessage(tvw.hwnd, TVM_GETEDITCONTROL, 0, 0)

    If lngHwnd <> 0

        SendMessage lngHwnd, EM_LIMITTEXT, intSize, 0

    End If 

End Sub
Probleme – Häufiger Fehler:

Die Item-Methode (Element), die eingesetzt werden kann, um ein bestimmtes Element der Nodes-Auflistung (Knotenauflistung) zurückzumelden, benötigt ein Argument. Handelt es sich bei diesem Argument um einen numerischen Ausdruck, sucht die Item-Methode auf Indexbasis nach dem Knoten; ist das Argument ein Zeichenfolgenausdruck, sucht die Item-Methode auf Schlüsselbasis nach dem Knoten. Da es sich bei diesem Argument um eine Variante handelt, kann die Item-Methode einen Schlüsselausdruck nicht von einem Indexausdruck unterscheiden, sofern der Ausdruck nicht mindestens ein numerisches Zeichen enthält. 


Lösung:
Verketten Sie mindestens ein numerisches Zeichen mit dem Ausdruck im Schlüsselargument der Add-Methode. Fügen Sie zum Beispiel einen Knoten hinzu, in dem Sie den Code 

Set Node = Me!CustOrders.Nodes.Add(, , "n" & rst!OrderID, Str(rst!OrderID))


anstelle des Codes 

Set Node = Me!CustOrders.Nodes.Add(, , rst!OrderID, CStr(rst!OrderID))


verwenden.

Siehe dazu MS-Artikel: D40088, D30952

